
22 Online Belief State Planning

Online methods determine the optimal policy by planning from the current belief
state. The belief space reachable from the current state is typically small compared
with the full belief space. As introduced in the fully observable context, many
online methods use variations of tree-based search up to some horizon.1 Various 1 A survey is provided by S.

Ross, J. Pineau, S. Paquet, and
B. Chaib-draa, “Online Planning
Algorithms for POMDPs,” Journal
of Artificial Intelligence Research,
vol. 32, pp. 663–704, 2008.

strategies can be used to try to avoid the exponential computational growth with
the tree depth. Although online methods require more computation per decision
step during execution than offline approaches, online methods are sometimes
easier to apply to high-dimensional problems.

22.1 Lookahead with Rollouts

Algorithm 9.1 introduced lookahead with rollouts as an online method in fully
observed problems. The algorithm can be used directly for partially observed
problems. It uses a function for randomly sampling the next state, which corre-
sponds to a belief state in the context of partial observability. This function was
already introduced in algorithm 21.11. Because we can use a generative model
rather than an explicit model for transitions, rewards, and observations, we can
accommodate problems with high-dimensional state and observation spaces.

22.2 Forward Search

b

...... ...... ...... ...... ...... ...... ...... ...

a(1) o(1) a(2) o(2)

Figure 22.1. Forward search
searches the action-observation-
belief graph to an arbitrary finite
depth in order to select the action
that produces the highest expected
reward. This illustration shows a
search to depth 2.

We can apply the forward search strategy from algorithm 9.2 to partially observed
problems without modification. The difference between MDPs and POMDPs is
encapsulated by one-step lookahead, which branches on actions and observations,
as shown in figure 22.1. The value of taking action a from belief b can be defined

recursively to a depth d:

Qd(b, a) =







R(b, a) + γ ∑o P(o | b, a)Ud−1(Update(b, a, o)) if d > 0

U(b) otherwise
(22.1)

where Ud(b) = maxa Qd(b, a). When d = 0, we have reached maximum depth
and return the utility using the approximate value function U(b), which may be
obtained from one of the methods discussed in the previous chapter, heuristically
chosen, or estimated from one or more rollouts. When d > 0, we continue to
search deeper, recursing down another level. Example 22.1 shows how to combine
QMDP with forward search for the machine replacement problem. Example 22.2
demonstrates forward search on the crying baby problem.

Consider applying forward search to the machine replacement problem.
We can first obtain an approximate value function through QMDP (algo-
rithm 21.2). We can then construct a ForwardSearch object, which was orig-
inally defined in algorithm 9.2. The call to lookahead within that function
will use the one defined for POMDPs in algorithm 20.5. The following code
applies forward search to the problem 𝒫 from belief state [0.5, 0.2, 0.3] to
depth 5 using our estimate of the utility obtained from QMDP at the leaf
nodes:
k_max = 10 # maximum number of iterations of QMDP
πQMDP = solve(QMDP(k_max), 𝒫)
d = 5 # depth
U(b) = utility(πQMDP, b)
π = ForwardSearch(𝒫, d, U)
π([0.5,0.2,0.3])

Example 22.1. Applying forward
search to the machine replacement
problem (appendix F.8).

The computation associated with the recursion in equation (22.1) grows expo-
nentially with depth, O(|A|d|O|d). Hence, forward search is generally limited to
a relatively shallow depth. To go deeper, we can limit the action or observation
branching. For example, if we have some domain knowledge, we may restrict the
action set either at the root or farther down the tree. For the observation branching,
we may restrict our consideration to a small set of likely observations—or even
just the most likely observation.2 Branching can be avoided entirely by adopting

2 R. Platt Jr., R. Tedrake, L. P. Kael-
bling, and T. Lozano-Pérez, “Belief
Space Planning Assuming Maxi-
mum Likelihood Observations,” in
Robotics: Science and Systems, 2010.

the open loop or hindsight optimization methods described in section 9.9.3 with
states sampled from the current belief.

22.2. forward search 455

Consider forward search with the crying baby problem with an approximate
value function given by the alpha vectors [−3.7,−15] and [−2,−21]. Running
forward search to depth 2 from the initial belief b = [0.5, 0.5] proceeds as
follows:

Q2(b, afeed) = R(b, afeed) + γ(P(crying | b, feed)U1([1.0, 0.0])

+ P(quiet | b, feed)U1([1.0, 0.0]))

= −10 + 0.9(0.1×−3.2157 + 0.9×−3.2157)

= −12.894

Q2(b, aignore) = R(b, aignore) + γ(P(crying | b, ignore)U1([0.093, 0.907])

+ P(quiet | b, ignore)U1([0.786, 0.214]))

= −5 + 0.9(0.485×−15.872 + 0.515×−7.779)

= −15.534

Q2(b, asing) = R(b, asing) + γ(P(crying | b, sing)U1([0.0, 1.0])

+ P(quiet | b, sing)U1([0.891, 0.109]))

= −5.5 + 0.9(0.495×−16.8 + 0.505×−5.543)

= −15.503

Recall that feeding the baby always results in a sated baby (b = [1, 0]), and
singing to the baby ensures that it cries only if it is hungry (b = [0, 1]).
Each U1 value is evaluated by recursing one level deeper in equa-
tion (22.1) using Ud(b) = maxa Qd(b, a). At maximum depth, we use
the approximate value function given by the alpha vectors, Q0(b, a) =

max
(

b⊤[−3.7,−15], b⊤[−2,−21]
).

The policy predicts that feeding the babywill result in the highest expected
utility, so it recommends that action.

Example 22.2. Forward search ap-
plied to the crying baby problem
(appendix F.7).

22.3 Branch and Bound

The branch and bound technique originally introduced in the context of MDPs can
be extended to POMDPs as well. The same algorithm in section 9.4 can be used
without modification (see example 22.3), relying on the appropriate lookahead
implementation to update beliefs and account for the observations. The efficiency
of the algorithm still depends on the quality of the upper and lower bounds for
pruning.

Although we can use domain-specific heuristics for the upper and lower
bounds, as we did in the fully observed case, we can alternatively use one of
the methods introduced in the previous chapter for discrete state spaces. For
example, we can use the fast informed bound for the upper bound and point-
based value iteration for the lower bound. So long as the lower bound U and
upper bound Q are true lower and upper bounds, the result of the branch and
bound algorithm will be the same as the forward search algorithm with U as the
approximate value function.

In this example, we apply branch and bound to the crying baby problem
with a depth of 5. The upper bound comes from the fast informed bound,
and the lower bound comes from point-based value iteration. We compute
the action from belief [0.4, 0.6] as follows:
k_max = 10 # maximum number of iterations for bounds
πFIB = solve(FastInformedBound(k_max), 𝒫)
d = 5 # depth
Uhi(b) = utility(πFIB, b)
Qhi(b,a) = lookahead(𝒫, Uhi, b, a)
B = [[p, 1 - p] for p in 0.0:0.2:1.0]
πPBVI = solve(PointBasedValueIteration(B, k_max), 𝒫)
Ulo(b) = utility(πPBVI, b)
π = BranchAndBound(𝒫, d, Ulo, Qhi)
π([0.4,0.6])

Example 22.3. An application of
branch and bound to the crying
baby problem.

22.4 Sparse Sampling

Forward search sums over all possible observations, resulting in a runtime ex-
ponential in |O|. As introduced in section 9.5, we can use sampling to avoid
exhaustive summation. We can generate m observations for each action and then

compute

Qd(b, a) =







1
m ∑

m
i=1

(

r
(i)
a + γUd−1

(

Update(b, a, o
(i)
a )
))

if d > 0

U(b) otherwise
(22.2)

where r
(i)
a and o

(i)
a are the ith-sampled observation and reward associated with

action a from belief b, and U(b) is the value function estimate at maximum depth.
We may use algorithm 9.4 without modification. The resulting complexity is
O(|A|dmd).

22.5 Monte Carlo Tree Search

The Monte Carlo tree search approach for MDPs can be extended to POMDPs,
though we cannot use the same exact implementation.3 The input to the algo- 3 Silver andVeness present aMonte

Carlo tree search algorithm for
POMDPs called Partially Observ-
ableMonte Carlo Planning (POMCP)
and show its convergence. D. Sil-
ver and J. Veness, “Monte-Carlo
Planning in Large POMDPs,” in
Advances in Neural Information Pro-
cessing Systems (NIPS), 2010.

rithm is a belief state b, depth d, exploration factor c, and rollout policy π.4 The

4 MonteCarlo tree search can be im-
plemented with a POMDP rollout
policy that operates on beliefs, or
on an MDP rollout policy that op-
erates on states. Random policies
are commonly used.

main difference between the POMDP algorithm (algorithm 22.1) and the MDP
algorithm is that the counts and values are associated with histories instead of
states. A history is a sequence of past actions and observations. For example, if
we have two actions a1 and a2 and two observations o1 and o2, then a possible
history could be the sequence h = a1o2a2o2a1o1. During the execution of the
algorithm, we update the value estimates Q(h, a) and counts N(h, a) for a set of
history-action pairs.5

5 There are many variations of the
basic algorithm introduced here,
including some that incorporate as-
pects of double progressive widen-
ing, discussed in section 9.6. Z.N.
Sunberg and M. J. Kochenderfer,
“Online Algorithms for POMDPs
with Continuous State, Action, and
Observation Spaces,” in Interna-
tional Conference on Automated Plan-
ning and Scheduling (ICAPS), 2018.

The histories associated with Q and N may be organized in a tree similar to
the one in figure 22.2. The root node represents the empty history starting from
the initial belief state b. During the execution of the algorithm, the tree structure
expands. The layers of the tree alternate between action nodes and observation
nodes. Associatedwith each action node are values Q(h, a) and N(h, a), where the
history is determined by the path from the root node. As with the MDP version,
when searching down the tree, the algorithm takes the action that maximizes

Q(h, a) + c

√

log N(h)

N(h, a)
(22.3)

where N(h) = ∑a N(h, a) is the total visit count for history h and c is an explo-
ration parameter. Importantly, c augments the value of actions that are unexplored
and underexplored, thus representing the relative trade-off between exploration
and exploitation.

struct HistoryMonteCarloTreeSearch
𝒫 # problem
N # visit counts
Q # action value estimates
d # depth
m # number of simulations
c # exploration constant
U # value function estimate

end

function explore(π::HistoryMonteCarloTreeSearch, h)
𝒜, N, Q, c = π.𝒫.𝒜, π.N, π.Q, π.c
Nh = sum(get(N, (h,a), 0) for a in 𝒜)
return argmax(a->Q[(h,a)] + c*bonus(N[(h,a)], Nh), 𝒜)

end

function simulate(π::HistoryMonteCarloTreeSearch, s, h, d)
if d ≤ 0

return π.U(s)
end
𝒫, N, Q, c = π.𝒫, π.N, π.Q, π.c
𝒮, 𝒜, TRO, γ = 𝒫.𝒮, 𝒫.𝒜, 𝒫.TRO, 𝒫.γ
if !haskey(N, (h, first(𝒜)))

for a in 𝒜
N[(h,a)] = 0
Q[(h,a)] = 0.0

end
return π.U(s)

end
a = explore(π, h)
s′, r, o = TRO(s,a)
q = r + γ*simulate(π, s′, vcat(h, (a,o)), d-1)
N[(h,a)] += 1
Q[(h,a)] += (q-Q[(h,a)])/N[(h,a)]
return q

end

function (π::HistoryMonteCarloTreeSearch)(b, h=[])
for i in 1:π.m

s = rand(SetCategorical(π.𝒫.𝒮, b))
simulate(π, s, h, π.d)

end
return argmax(a->π.Q[(h,a)], π.𝒫.𝒜)

end

Algorithm 22.1. Monte Carlo tree
search for POMDPs from belief
b. The initial history h is optional.
This implementation is similar to
the one in algorithm 9.5.

Figure 22.2. A search tree con-
taining all the histories covered
when running a Monte Carlo tree
search with 100 samples on the ma-
chine replacement problem. Visi-
tations are given beneath each ac-
tion node, and color indicates node
values with high values in blue
and low values in red. Expanded
nodes with zero visitations are not
shown. This search used an explo-
ration constant c = 0.5, a max-
imum depth d = 5, and a uni-
form random rollout policy. The
initial belief is certainty in a fully
working system. Monte Carlo tree
search is able to avoid certain ac-
tions and instead focus samples on
more promising paths.As with the MDP version, the Monte Carlo tree search algorithm is an anytime

algorithm. The loop in algorithm 22.1 can be terminated at any time, and the
best solution found up to that point will be returned. With a sufficient number of
iterations, the algorithm converges to the optimal action.

Prior knowledge can be incorporated into Monte Carlo tree search in how we
initialize N and Q. Our implementation uses zero, but other choices are possible,
including having the initialization of the action values be a function of history.
The value estimates can again be obtained from simulations of a rollout policy.

The algorithm does not need to be reinitialized with each decision. The history
tree and associated counts and value estimates can be maintained between calls.
The observation node associated with the selected action and actual observation
becomes the root node at the next time step.

22.6 Determinized Sparse Tree Search

Determinized sparse tree search strives to reduce the overall amount of sampling in
both sparse sampling and Monte Carlo tree search by making the observation
resulting from performing an action deterministic.6 It does so by building a

6 Ye, Somani, Hsu, and Lee present
a determinized sparse tree search
algorithm for POMDPs called De-
terminized Sparse Partially Observ-
able Tree (DESPOT) N. Ye, A.
Somani, D. Hsu, and W. S. Lee,
“DESPOT: Online POMDP Plan-
ning with Regularization,” Jour-
nal of Artificial Intelligence Research,
vol. 58, pp. 231–266, 2017. In addi-
tion, the algorithm includes branch
and bound, heuristic search, and
regularization techniques.

determinized belief tree from a special particle belief representation to form a sparse
approximation of the true belief tree. Each particle refers to one of m scenarios,
each of depth d. A scenario represents a fixed history that the particle will follow
for any given sequence of actions a(1), a(2), . . . , a(d). Every distinct action sequence

460 chapter 22. online belief state planning

produces a distinct history under a particular scenario.7 This determinization 7 A similar idea was discussed in
section 9.9.3 and is related to the
PEGASUS algorithm mentioned in
section 11.1.

reduces the size of the search tree to O(|A|dm). An example of a history is given
in example 22.4. A determinized tree is shown in figure 22.3.

Suppose we have two states s1 and s2, two actions a1 and a2, and two ob-
servations o1 and o2. A possible history of depth d = 2 for the particle with
initial state s2 is the sequence h = s2a1o2s1a2o1. If this history is used as a
scenario, then this history is returned every time the belief tree is traversed
from the initial state with the action sequence a(1) = a1 and a(2) = a2.

Example 22.4. A history and a
scenario in the context of deter-
minized sparse tree search.

a(1) o(1) a(2) o(2)

b

Figure 22.3. A determinized
sparse search tree with two scenar-
ios, shown in blue and purple. The
line traces show the possible paths
for each scenario under different
action sequences.

A search tree with m scenarios up to depth d can be fully specified by a compact
m × d determinizing matrix Φ containing probability masses. The element Φij

contains the information needed for a particle following the ith scenario at depth
j to identify its successor state and observation. Specifically, Φij is a uniformly
distributed randomnumber that can generate the successor pair (s′, o) froma state-
action pair (s, a), following the distribution P(s′, o | s, a) = T(s′ | s, a)O(o | a, s′).
We can generate a determinizingmatrix by filling itwith values sampleduniformly
between 0 and 1.

Beliefs are represented as vectors of belief particles. Each belief particle φ con-
tains a state s and indices i and j into the determinizing matrix Φ corresponding
to a scenario i and current depth j. Given a particular action a, Φij is used to
deterministically transition to successor state s′ and observation o. The successor
particle φ′ = (s′, i, j + 1) receives s′ as its state and increments j by 1. Exam-
ple 22.5 demonstrates this tree traversal process. The particle belief representation
is implemented in algorithm 22.2 and is used in forward search in algorithm 22.3.

22.7 Gap Heuristic Search

Similar to the offline heuristic search presented in section 21.8, gap heuristic search
uses the gap between the upper and lower bounds to guide our search toward
beliefs that have uncertainty in their associated value and as an indication of
when we can stop exploring. The gap at a belief b is the difference between the
upper-bound and lower-bound values: U(b)−U(b). Search algorithms with the
gap heuristic select the observation that maximizes the gap because they are more
likely to benefit from a belief backup. Actions are often selected according to a

Suppose that we generate a determinizing matrix Φ for a 3-state,
2-observation problem with four histories up to depth 3:

Φ =











0.393 0.056 0.369

0.313 0.749 0.273

0.078 0.262 0.009

0.969 0.598 0.095











Suppose that we take action a3 in state s2 when at depth 2 while following
history 3. The corresponding belief particle is φ = (2, 3, 2), and the deter-
minizing value in Φ is Φ3,2 = 0.262.

The deterministic successor action and observation are given by iterating
over all successor state-observation pairs and accumulating their transition
probabilities. We begin with p = 0 and evaluate s′ = s1, o = o1. Suppose we
get T(s1 | s2, a3)O(o1 | a3, s1) = 0.1. We increase p to 0.1, which is less than
Φ3,2, so we continue.

Next, we evaluate s′ = s2, o = o1. Suppose we get T(s2 | s2, a3)O(o1 |
a3, s2) = 0.17. We increase p to 0.27, which is greater than Φ3,2. We thus
deterministically proceed to s′ = s2, o = o1 as our successor state, resulting
in a new particle φ′ = (1, 3, 3).

Example 22.5. Determinized
sparse tree search uses a matrix to
make tree traversal deterministic
for a given particle.

struct DeterminizedParticle
s # state
i # scenario index
j # depth index

end

function successor(𝒫, Φ, ϕ, a)
𝒮, 𝒪, T, O = 𝒫.𝒮, 𝒫.𝒪, 𝒫.T, 𝒫.O
p = 0.0
for (s′, o) in product(𝒮, 𝒪)

p += T(ϕ.s, a, s′) * O(a, s′, o)
if p ≥ Φ[ϕ.i, ϕ.j]

return (s′, o)
end

end
return last(𝒮), last(𝒪)

end

function possible_observations(𝒫, Φ, b, a)
𝒪 = []
for ϕ in b

s′, o = successor(𝒫, Φ, ϕ, a)
push!(𝒪, o)

end
return unique(𝒪)

end

function update(b, Φ, 𝒫, a, o)
b′ = []
for ϕ in b

s′, o′ = successor(𝒫, Φ, ϕ, a)
if o == o′

push!(b′, DeterminizedParticle(s′, ϕ.i, ϕ.j + 1))
end

end
return b′

end

Algorithm 22.2. The determinized
particle belief update used in de-
terminized sparse tree search for a
POMDP 𝒫. Each belief b consists of
particles ϕ that each encode a par-
ticular scenario and depth along
the scenario. Their scenario’s tra-
jectory is determinized through a
matrix Φ containing random values
in [0, 1]. Each particle ϕ represents
a particular scenario i at a particu-
lar depth j, referring to the ith row
and jth column of Φ.

struct DeterminizedSparseTreeSearch
𝒫 # problem
d # depth
Φ # m×d determinizing matrix
U # value function to use at leaf nodes

end

function determinized_sparse_tree_search(𝒫, b, d, Φ, U)
𝒮, 𝒜, 𝒪, T, R, O, γ = 𝒫.𝒮, 𝒫.𝒜, 𝒫.𝒪, 𝒫.T, 𝒫.R, 𝒫.O, 𝒫.γ
if d == 0

return (a=nothing, u=U(b))
end
best = (a=nothing, u=-Inf)
for a in 𝒜

u = sum(R(ϕ.s, a) for ϕ in b) / length(b)
for o in possible_observations(𝒫, Φ, b, a)

Poba = sum(sum(O(a,s′,o)*T(ϕ.s,a,s′) for s′ in 𝒮)
for ϕ in b) / length(b)

b′ = update(b, Φ, 𝒫, a, o)
u′ = determinized_sparse_tree_search(𝒫,b′,d-1,Φ,U).u
u += γ*Poba*u′

end
if u > best.u

best = (a=a, u=u)
end

end
return best

end

function determinized_belief(b, 𝒫, m)
particles = []
for i in 1:m

s = rand(SetCategorical(𝒫.𝒮, b))
push!(particles, DeterminizedParticle(s, i, 1))

end
return particles

end

function (π::DeterminizedSparseTreeSearch)(b)
particles = determinized_belief(b, π.𝒫, size(π.Φ,1))
return determinized_sparse_tree_search(π.𝒫,particles,π.d,π.Φ,π.U).a

end

Algorithm 22.3. An implemen-
tation of determinized sparse
tree search, a modification of
forward search, for POMDPs.
The policy takes a belief b in the
form of a vector of probabilities,
which is approximated by a vector
of determinized particles by
determinized_belief.

© 2022 Massachusetts Institute of Technology, shared under a Creative Commons CC-BY-NC-ND license.
2024-02-06 20:54:49-08:00, comments to bugs@algorithmsbook.com



464 chapter 22. online belief state planning

lookahead using an approximate value function. Algorithm 22.4 provides an
implementation.8 8 There are a variety of differ-

ent heuristic search algorithms for
POMDPs that attempt to mini-
mize the gap. For example, see S.
Ross and B. Chaib-draa, “AEMS:
An Anytime Online Search Algo-
rithm for Approximate Policy Re-
finement in Large POMDPs,” in In-
ternational Joint Conference on Artifi-
cial Intelligence (IJCAI), 2007. This
implementation is similar to the
one used by DESPOT, referenced
in the previous section.

The initial lower- and upper-bound values used in heuristic search play an
important role in the algorithm’s performance. Example 22.6 uses a random
rollout policy for the lower bound U(b). A rollout is not guaranteed to produce a
lower bound, of course, because it is based on a single trial up to a fixed depth.
As the number of samples increases, it will converge to a true lower bound. That
example uses the best-action best-state upper bound from equation (21.2). Many
other forms of upper and lower bounds exist that can provide faster convergence,
but at the cost of run time and implementation complexity. For example, using
the fast informed bound (algorithm 21.3) for the upper bound can improve
exploration and help reduce the gap. For the lower bound, we can use a problem-
specific rollout policy to better guide the search.

22.8 Summary

• A simple online strategy is to perform a one-step lookahead, which considers
each action taken from the current belief and estimates its expected value using
an approximate value function.

• Forward search is a generalization of lookahead to arbitrary horizons, which
can lead to better policies, but its computational complexity grows exponen-
tially with the horizon.

• Branch and bound is a more efficient version of forward search that can avoid
searching certain paths by placing upper and lower bounds on the value
function.

• Sparse sampling is an approximation method that can reduce the computa-
tional burden of iterating over the space of all possible observations.

• Monte Carlo tree search can be adapted to POMDPs by operating over histories
rather than states.

• Determinized sparse tree search uses a special form of particle belief that
ensures that observations are determinized, greatly reducing the search tree.

struct GapHeuristicSearch
𝒫 # problem
Ulo # lower bound on value function
Uhi # upper bound on value function
δ # gap threshold
k_max # maximum number of simulations
d_max # maximum depth

end

function heuristic_search(π::GapHeuristicSearch, Ulo, Uhi, b, d)
𝒫, δ = π.𝒫, π.δ
𝒮, 𝒜, 𝒪, R, γ = 𝒫.𝒮, 𝒫.𝒜, 𝒫.𝒪, 𝒫.R, 𝒫.γ
B = Dict((a,o)=>update(b,𝒫,a,o) for (a,o) in product(𝒜,𝒪))
B = merge(B, Dict(()=>copy(b)))
for (ao, b′) in B

if !haskey(Uhi, b′)
Ulo[b′], Uhi[b′] = π.Ulo(b′), π.Uhi(b′)

end
end
if d == 0 || Uhi[b] - Ulo[b] ≤ δ

return
end
a = argmax(a -> lookahead(𝒫,b′->Uhi[b′],b,a), 𝒜)
o = argmax(o -> Uhi[B[(a, o)]] - Ulo[B[(a, o)]], 𝒪)
b′ = update(b,𝒫,a,o)
heuristic_search(π,Ulo,Uhi,b′,d-1)
Ulo[b] = maximum(lookahead(𝒫,b′->Ulo[b′],b,a) for a in 𝒜)
Uhi[b] = maximum(lookahead(𝒫,b′->Uhi[b′],b,a) for a in 𝒜)

end

function (π::GapHeuristicSearch)(b)
𝒫, k_max, d_max, δ = π.𝒫, π.k_max, π.d_max, π.δ
Ulo = Dict{Vector{Float64}, Float64}()
Uhi = Dict{Vector{Float64}, Float64}()
for i in 1:k_max

heuristic_search(π, Ulo, Uhi, b, d_max)
if Uhi[b] - Ulo[b] < δ

break
end

end
return argmax(a -> lookahead(𝒫,b′->Ulo[b′],b,a), 𝒫.𝒜)

end

Algorithm 22.4. An implementa-
tion of heuristic search that uses
bounds, a gap criterion, and ini-
tial lower and upper bounds on
the value function. We update a
dictionary Ulo and Uhi to main-
tain the lower and upper bounds
on the value function as spe-
cific beliefs. At belief b, the gap
is Uhi[b] - Ulo[b]. Exploration
stops when the gap is smaller than
the threshold δ or the maximum
depth d_max is reached. A maxi-
mum number of iterations k_max
is allotted to search.

466 chapter 22. online belief state planning

The following code demonstrates how to apply gap heuristic search to the
crying baby problem.
δ = 0.001 # gap threshold
k_max = 5 # maximum number of iterations
d_max = 10 # maximum depth
πrollout(b) = rand(𝒜) # random rollout policy
Ulo(b) = rollout(𝒫, b, πrollout, d_max) # initial lower bound
Rmax = maximum(R(s,a) for (s,a) in product(𝒮,𝒜)) # max reward
Uhi(b) = Rmax / (1.0 - 𝒫.γ) # best action best state upper bound
π = GapHeuristicSearch(𝒫, Ulo, Uhi, δ, k_max, d_max)
π([0.5, 0.5]) # evaluate at initial belief point

Here, we show six iterations of heuristic search with an initial belief b of
[0.5, 0.5]. In each iteration, the upper bound is shown in green and the lower
bound is shown in blue.

The jagged bounds are due to some beliefs not being reexplored based on
the action and observation selection. In the bottom row, we see that it has
explored many of the beliefs once, but the bounds are still loose. Heuristic
search seeks to reduce the maximum gap.

Example 22.6. The use of gap
heuristic search lower and upper
bounds for the crying baby prob-
lem over iterations of heuristic
search.

• Heuristic search intelligently selects action-observation pairs to explore regions
with a high gap between the upper and lower bounds on the value function
that it maintains.

22.9 Exercises

Exercise 22.1. Suppose we have A = {a1, a2} and a belief b = [0.5, 0.5]. The reward is
always 1. The observation function is given by P(o1 | a1) = 0.8 and P(o1 | a2) = 0.4. We
have an approximate value function represented by an alpha vector α = [−3, 4]. With
γ = 0.9, use forward search to a depth of 1 to compute U(b). Use the following updated
beliefs in the calculation:

a o Update(b, a, o)

a1 o1 [0.3, 0.7]

a2 o1 [0.2, 0.8]

a1 o2 [0.5, 0.5]

a2 o2 [0.8, 0.2]

Solution: We need to calculate the action value function at depth 1 according to equa-
tion (22.1):

Qd(b, a) = R(b, a) + γ ∑
o

P(o | b, a)Ud−1(Update(b, a, o))

First, we calculate the utility for the updated beliefs:

U0(Update(b, a1, o1)) = α⊤b′ = 0.3×−3 + 0.7× 4 = 1.9

U0(Update(b, a2, o1)) = 0.2×−3 + 0.8× 4 = 2.6

U0(Update(b, a1, o2)) = 0.5×−3 + 0.5× 4 = 0.5

U0(Update(b, a2, o2)) = 0.8×−3 + 0.2× 4 = −1.6

Second, we compute the action value function for both actions:

Q1(b, a1) = 1 + 0.9((P(o1 | b, a1)U0(Update(b, a1, o1)) + (P(o2 | b, a1)U0(Update(b, a1, o2)))

= 1 + 0.9(0.8× 1.9 + 0.2× 0.5) = 2.458

Q1(b, a2) = 1 + 0.9((P(o1 | b, a2)U0(Update(b, a2, o1)) + (P(o2 | b, a2)U0(Update(b, a2, o2)))

= 1 + 0.9(0.4× 2.6 + 0.6×−1.6) = 1.072

Finally, we have U1(b) = maxa Q1(b, a) = 2.458.

Exercise 22.2. Using the following trajectory samples, compute the action value function
for belief b and actions a1 and a2 based on sparse sampling to depth 1. Use the following
updated beliefs, discount factor γ = 0.9, and approximate value function represented by
an alpha vector α = [10, 1].

a o r Update(b, a, o)

1 1 0 [0.47, 0.53]

2 1 1 [0.22, 0.78]

1 2 1 [0.49, 0.51]

2 1 1 [0.22, 0.78]

2 2 1 [0.32, 0.68]

1 2 1 [0.49, 0.51]

Solution: We first calculate the utility for the updated beliefs:

a o r Update(b, a, oa) U0(Update(b, a, o))

1 1 0 [0.47, 0.53] 5.23

2 1 1 [0.22, 0.78] 2.98

1 2 1 [0.49, 0.51] 5.41

2 1 1 [0.22, 0.78] 2.98

2 2 1 [0.32, 0.68] 3.88

1 2 1 [0.49, 0.51] 5.41

Then, we can compute the action value function over all actions using equation (22.2):

Q1(b, a1) =
1

3
(0 + 1 + 1 + 0.9(5.23 + 5.41 + 5.41)) = 5.48

Q1(b, a2) =
1

3
(1 + 1 + 1 + 0.9(2.98 + 2.98 + 3.88)) = 3.95

Exercise 22.3. Consider example 22.5. Suppose we have the following transition functions:
T(s2 | s1, a3) = 0.4

T(s3 | s1, a3) = 0.45

O(o1 | s2, a3) = 0.6

O(o2 | s2, a3) = 0.5

O(o1 | s3, a3) = 0.4

O(o2 | s3, a3) = 0.1

What is the path taken by a particle associated with φ = (1, 4, 2) if we take action a3?
Solution: From the determinizing matrix, our determinizing value is Φ4,2 = 0.598 and we
are in state s1. Then, we calculate p as follows:

p← T(s2 | s1, a3)O(o1 | s2, a3) = 0.4× 0.6 = 0.24

p← p + T(s3 | s1, a3)O(o1 | s3, a3) = 0.24 + 0.45× 0.4 = 0.42

p← p + T(s2 | s1, a3)O(o2 | s2, a3) = 0.42 + 0.4× 0.5 = 0.67

We stop our iteration because p > 0.598. Thus, from our final iteration, we proceed to
(s2, o2).

Exercise 22.4. Summarize the techniques covered in this chapter to reduce branching over
actions.

Solution: Branch and bound can reduce action branching by using an upper bound on the
value function. It skips actions that cannot improve on the value obtained from actions that
it explored earlier. Gap heuristic search and Monte Carlo tree search use approximations
of action values to guide the selection of actions during exploration.

Exercise 22.5. Summarize the techniques covered in this chapter to reduce branching over
observations.

Solution: Sparse sampling reduces observation branching by sampling only a small number
of observations. Observations are sampled from P(o | b, a), which means that observations
that have greater probability are more likely to be sampled. Determinized sparse tree
search uses a similar approach, but the sampling occurs once and is then fixed. Branching
over observations can also be reduced based on the lookahead value U(b′). Gap heuristic
search evaluates the gap and avoids branching on observations for which we have high
confidence in the value function.

